Introduction to Mplus: Brief Version
Links of Importance

(1) The homepage for Mplus is www.statmodel.com
(2) You can download the free version of the software at www.statmodel.com/demo.shtml
(3) The user's manual, which you will want to have if you plan to run any programs using Mplus, is at www.statmodel.com/ugexerpts.shtml
(4) Also, some of the technical information about the principles behind the software's operations can be found at www.statmodel.com/techappen.shtml
(5) Sometimes it can be helpful to look for answers to questions at www.statmodel.com/cgi-bin/discus/discus.cgi
If you purchase the software and keep your annual renewal up, you can e-mail them questions, which usually receive prompt replies in my experience.

Why Mplus?

There can be alot of stress over the selection of the right software and there are some very good programs available. I will not provide a comparative review of all the packages, just the top two things I like about Mplus. 

(1) Features: Mplus has many great features. I am of the opinion that at the present time, Mplus cannot be beat for features for the commercial SEM packages.

(2) Expected Future Development: The Mplus team of Linda and Bengt Muthen has been very impressive in their development and continuous support of their software in the years since it was launched (actually, it evolved from an earlier package). Bengt, as of fall 2006, took early retirement from UCLA to join Linda and the rest of the team in working full time on the software and on training. Their week-long class is very deep, providing the student with an amazing array of tools for use in analyses. 

(3) Cost: Mplus is a commercial package, and therefore has a nontrivial cost. There is also a nontrival annual fee to qualify for updates as they are developed.

(4) The Language: The command structure for Mplus makes this package less intuitive for modeling. One needs to develop a graph of their model separately in many cases to keep track of their models. The condensed nature of the code, along with the form of the user’s manual, make Mplus a difficult program to use for learning about SEM. However, the tutorials in this website should help in that regard to some degree.
Mplus Overview
Mplus is a command-based program. Its capabilities are based around 10 command statements:

TITLE:

DATA:

VARIABLE:

DEFINE:

ANALYSIS:

MODEL:

OUTPUT:

SAVEDATA:

PLOT:

MONTECARLO:

A few points:

(1) The DATA and VARIABLE commands are the only two that are required. 
(2) This is a good place to mention that the command language is set up with defaults so as to minimize the amount of text you have to type. Thus, it is important to get to know the defaults, particularly for the MODEL statement, which is the most complex part of the program. 
(3) It is also important to know that the individual lines of code cannot exceed 80 characters. 
(4) The command words IS, ARE, and = are interchangable anywhere except in the DEFINE command. Items in a list (e.g. variable names) can be separated either by blanks or commas.
(5)  Here is an example program.
TITLE:           Example 1
DATA:           FILE IS example1.dat;  !exclamations allow comments on lines

VARIABLE:  NAMES ARE x1 x2 x3 y1 y2;

MODEL:        y1 ON x1 x2 x3;  !the keyword ON means "regressed on" (aka =)

                       y2 ON x1 y1;

The above five lines are all you would need to run the path model below with three exogenous (independent) variables and two response variables.

Basic Data Format

There are several options for working with summary data, as well as data with special formats. The basic data format is a text file, with rows containing records and columns containing variables. All data must be numeric (no alpha characters) and variables names are not included in the data file, but specified in the program. If you are working from an Excel file, for example, the simplest way to create the text file maybe to simply copy and paste from the Excel file into an empty Mplus window. When you save that file, you will want to select the file type option "all files" rather than the default ".inp". 
Command Statement Options

Additional information about the option statements associated with various commands can be found in the descriptions below. Note that only the first four letters of a command option statement are required in most cases (i.e., NOBSERVATIONS = NOBS), with exceptions, such as FULLCOV, FULLCORR, and MEANS. You are always safe using the full statement or the abbreviations I use. See user's manual for more details (if you have not already downloaded it, it can be found at www.statmodel.com/ugexerpts.shtml).

DATA Command
Command Options for DATA: 

DATA: 

     FILE IS                            file name;

     FORMAT IS                    format statement;

     TYPE IS                           COVARIANCE, CORRELATION, FULLCOV,

                                               FULLCORR, MEANS, STDEVIATIONS;

     NOBS ARE                      number of observations;

     NGROUPS =                    number of groups;

Note that the DATA command is required, as is the FILE option. 

FILE IS     

It is assumed that the named data file is in the same directory as the program file, unless a path is given. 

FORMAT IS

Normally data are in free format in a text file, with each row being a case and each column being a variable. Only alphanumeric values can be in the source data file, which means that nominal variables need to be numerically coded, usually as  a dummy (0, 1) variable. Format statements  can be specified using a FORTRAN-like statement like FORMAT IS 5F4.0, 10x, 6F1.0; See user's manual for more details.

TYPE IS

In addition to raw data files, there are several kinds of summary files that can be read by Mplus. These include covariance matrices, correlation matrices, means, and standard errors. Covariance and correlation matrices can be either half matrices or full. Mplus is able to perform appropriate analyses on correlation matrices because of the way in which the underlying implementation is done. This is not (or, at least, was not) the case for all software packages. This is relevant if you only have access to the correlations, for example, if you are working from published data summaries. Typically, however, you should be aware that it is nearly always preferable to use the covariances instead of just the correlations (unless you are teaching, in which correlations are useful to show). If you have both the correlations and the standard deviations, you have all the information you need for Mplus to calculate the covariances and conduct its analyses on them.  Mplus allows you to read the data as a separate correlation matrix plus the means and standard deviations. In this case, the data file should present the means first, the standard deviations second, and the correlations last. When all three of these are presented, the Command Option would be TYPE IS CORR MEANS STD; The data file might look like this: 

	rowtype_
	varname_
	stems00
	change
	nig00
	nig01
	lac00
	lac01

	n
	
	162
	162
	162
	162
	162
	162

	corr
	stems00
	1
	
	
	
	
	

	corr
	change
	-0.63
	1
	
	
	
	

	corr
	nig00
	0.51
	-0.28
	1
	
	
	

	corr
	nig01
	0.52
	-0.2
	0.75
	1
	
	

	corr
	lac00
	0.15
	-0.24
	-0.12
	-0.18
	1
	

	corr
	lac01
	0.31
	-0.39
	0.03
	0.03
	0.7
	1

	stddev
	
	1.1276
	18.031
	1.7045
	1.6647
	1.9756
	1.8714


example data file containing correlations and standard deviations.
NOBS =
Here we simply have to state the number of observations from which a summary data set was derived, since that information is not inherent in the file itself. So, we might say NOOBS ARE  105; if there were 105 rows of data in the original data set.

NGROUPS = 

When performing what is referred to as a "multigroup" analysis, which seeks to compare the models for two related datasets (e.g., males and females), we invoke the multigroup operations by declaring more than one group. So, if we include the Command Option NGROUPS = 3; the syntax reader knows how to interpret the MODEL command properly.

VARIABLE Command

Command Options for VARIABLE:

VARIABLE:

     NAMES =                          names of variables in data set;

     USEOBS =                         conditional statement about obs used in analysis;

     USEVAR =                        names of variables used in model that follows; 

     MISSING =                       .; (tells the program that . is symbol for missing value)

     CATEGORICAL =           names of categorical response variables;

     GROUPING =                   name of group variable;

     CLUSTER =                       name of cluster variable;

Much information of importance is conveyed in the VARIABLE command and there are many options. I will present only a few critical pieces of information here, the reader should refer to the user's manual for the rest.

USEOBS = 

Only names in the original VARIABLES ARE list can be used here (not new ones created using the DEFINE command). Typically, we would use logical operators to specify which observations to use. Logical operators are discussed under DEFINE (see left button). An example would be USEOBS = diameter GE 0.1 AND alive = 1;

USEVAR =

One of the most important things to know  is that if new variables are created using the DEFINE command (below), they must follow the original variables in the USEVAR = list. This does not apply unless a new variable name is created in the process (e.g., whiskers = whiskers/100; does not create a new variable name).

MISSING =

Note that there is no default symbol for missing values. Often people use a naked period ( . ) in a data set to represent missing values. Whatever is used should be specified. Be aware that the default in Mplus is listwise deletion, which means if a single value is missing for the variables measured on a case, that case is dropped from the analysis. There are many options for how to handle missing data (refer to user's manual).

Most of the other commands simply refer to declaring one of the variables as being of special significance, for example, as a code for group membership.

Define:

CATEGORICAL =
This refers ONLY to response (y) variables that are categorical. Note that exogenous categorical variables require no special treatment and should NOT be declared in the categorical statement. By declaring a variable to be categorical, Mplus treats is using a probit formulation unless a logistic form is requested.

GROUPING =

When performing multi-group modeling, you will need to declare which variable in the dataset specifies the groups.

CLUSTER = 

When cluster sampling has been used, the name of the cluster variable can be used to invoke a correction of the standard errors.
DEFINE Command

Command Options for DEFINE:

This command allows the user to create new variables. Most of the statements permitted here will be familiar. There are, however, a few rules. First, any variable in the input file, plus those created earlier in the DEFINE list, can be used in a statement. When you transform an existing variable with a statement like 

whiskers = whiskers/100;

there is no special actions required. However, if you create a new variable name, such as with

whisksqr = whiskers*whiskers;

then you need to add whisksqr to the USEVARS ARE list and the new variable names must follow those that are in the original data set. Of course, you only need to include a variable (original or defined) in the USEVARS ARE list if you are using that variable in the MODEL statement or as a special action variable, such as in the CLUSTER IS  statement.

Types of DEFINE Statements Available

There are three types of commands available:

     variable = mathematical expression;

     IF (conditional statement) THEN transformation statement;

     CUT variable name or list of variables (cutpoints);

Let's begin with the CUT option first, since it is the most mysterious. The CUT option is used to convert a continuous variable into an ordered categorial one. By using the statement

CUT y1 (50);

the continuous range of values for y1 are now converted into a set of 0 and 1 values, with those values less than or equal to 50 being turned into zeros and those original values greater than 50 being turned into ones. The statement

CUT y1 (25, 50, 75);

would convert the continuous values of variable y1 into one of four values, 0, 1, 2, or 3, depending on how their values compared to the specified cutpoints.

Logical Operators, Arithmetic Operators, and Functions

Now, for the variable transformations, we need to be familiar with the logical and arithmetic operators, as well as functions, in Mplus. These can be used in several different places in Mplus commands, such as in the USEOBS = statement (aka the USEOBS ARE or USEOBS IS statements). Note, however, that in the DEFINE command, ARE, IS, and = are not necessarily interchangable. 

Logical Operators:
The following are the logical operators and, if there are any, their equivalents.

AND     

OR

NOT

EQ     logical equal

NE     logical not equal

GE     logical greater than or equal,  >=

LE     logical less than or equal, <=

GT     greater than, >

LT     less than, <

Arithmetic Operators:
These are the usual operators for the most part.

+ is addition, - is subtraction, * is multiplication, / is division, ** is exponentiation. 

The less familiar operator is % is to obtain the remainder from division.

Functions

LOG       base e log

LOG10  base 10 log

EXP       exponential

SQRT    square root

ABS        absolute value

SIN         sine

COS        cosine

TAN       tangent

ASIN      arcsine

ACOS      arccosine

ATAN     arctangent

Conditional Statements

IF (condition) THEN newvar = value;

ANALYSIS Command

Command Options for ANALYSIS:

The ANALYSIS command is used to specify the type of analysis, the statistical estimation procedure, the matrix to be analyzed, and the specific algorithms to be used. The options follow. Don't worry too much about the vastness of the list. In most cases you will likely be using the defaults and when you are not, helpful error messages will steer you in the right direction. Still, it is useful to look at the manual to learn more about the available options. 

ANALYSIS:

     TYPE =           BASIC;

                                 GENERAL;

                                         MEANSTRUCTURE;

                                         MISSING;

                                         H1;

                                         MCOHORT;

                                         MIXTURE;

                                         COMPLEX;

                                         TWOLEVEL;

                                 EFA # #;

                                 LOGISTIC;

     ESTIMATOR =   ML;

                                         MLM;

                                         MLMV;

                                         MLR;

                                         MLF;

                                         WLS;

                                         WLSM;

                                         WLSMV;

                                         GLS;

                                         ULS;

     MATRIX =            COVARIANCE;

                                         CORRELATION;

     BOOTSTRAP = # of bootstrap draws;

     ITERATIONS =

     H1ITERATIONS =

     MITERATIONS =

     MCITERATIONS = 

     MUITERATIONS = 

     CONVERGENCE =  

     H1CONVERGENCE = 

     COVERAGE = 

     LOGCRITERION =

     MCONVERGENCE =

     MCCONVERGENCE = 

     MUCONVERGENCE = 

     MIXC =

     MIXU = 

     LOGHIGH = 

     LOGLOW =

     UCELLSIZE =

Analysis TYPES

There are four basic types of analyses, (1) BASIC, GENERAL, EFA, and LOGISTIC. GENERAL is the default and what is used in the great majority of cases. BASIC allows users to obtain sample statistics and other information about their data without estimating a model. EFA invokes exploratory factor analysis, which is not part of structural equation modeling, but which can be helpful when trying to decide what model to build for a set of data. LOGISTIC provides logistic regression results for models containing categorical response variables. Note that in Mplus, the analysis of categorical responses is usually based on a probit model instead of a logit model and is conducted within the GENERAL analysis type.

There are numerous options within the GENERAL analysis. These cover the full range of SEM capabilities, including growth modeling and multilevel modeling. 
MEANSTRUCTURE is used to analyze means, thresholds, and intercepts. MISSING is used to request the analysis of cases with missing values. H1 permits the estimation of an unrestricted means and covariance model when TYPE = MISSING. MCOHORT is used when analyzing multiple cohort data. MIXTURE allows for the inclusion of categorical latent variables in the model. COMPLEX is for data collected with a complex sample design (e.g., nested or blocked sampling). TWOLEVEL permits the simultaneous analysis of two levels of data. 

ESTIMATOR Option

The default estimator depends on the type of analysis and the variable scales. Only three types of estimation can be run with summary data (e.g., covariance matrix) and these are ML, GLS, and ULS. 

ML - Standard maximum likelihood estimation.

MLM - Maximum likelihood with robust standard error and a mean-adjusted chi-square.

MLMV - Maximum likelihood with robust standard error and a mean- and variance-adjusted chi-square.

MLR - Maximum likelihood with robust standard errors. 

MLF - Maximum likelihood with standard errors estimated using first-order derivatives.

WLS - Weighted least squares.

WLSM - Weighted least squares with robust standard errors and mean-adjusted chi-square.

WLSMV - Weighted least squares with robust standard errors and mean- and variance-adjusted chi-square.

GLS - Generalized least squares.

ULS - Unweighted least squares. There are no standard errors or chi-square test statistic with this analysis.

MATRIX Options

The default is to analyze the covariance matrix. The statement MATRIX = CORRELATION; The analysis of correlation matrices is only permitted when all dependent variables are continuous, there is no mean structure, and you are not performing a multigroup analysis. For models with all categorical dependent variables, the correlation matrix is used. If any dependent variables are continuous, the covariance matrix is analyzed. 

BOOTSTRAP Option

The declaration of the bootstrap option and the number of bootstrap draws activates the bootstrapping procedures. Bootstrapping requires that you are working with the raw data. You cannot use bootstrapping along with the robust soluation methods (MLF, MLM), as both give the same results. Also, residual bootstrapping (the Bollen-Stine method) is available only for continuous outcomes. When used alone, the BOOTSTRAP command results in alternative estimation of parameter standard errors and also for the chi-square p-value (residual bootstrapping).

ITERATIONS and CONVERGENCE Options

ITERATIONS specifies maximum iterations for solving standard models. 

H1ITERATIONS specifies iterations for models with missing data.

MITERATIONS specifies iterations for latent variable mixture models.

MCITERATIONS specifies number of M-step iterations for latent class regression part of latent variable mixture models. 

MUITERATIONS specifies number of M-step iterations for latent class indicator part of latent variable mixture models.

CONVERGENCE specifies the convergence criterion for all cases except the H1 model and the latent variable mixture model.

The rest of the above options are pretty specialized. See user's manual for information.

MODEL:

MODEL Command

Command Options for MODEL:

MODEL:

          BY, ON, WITH, PWITH  ! specifies relations among variables

          *   ! assigns starting values or frees a parameter for estimation

          @  ! used to fix a parameter value

          (number)  ! this syntax is used to constrain parameters to be equal

          list of variables;  ! refers to variances and residual variances

          [list of variables];  ! refers to means, intercepts, thresholds

          {list of variables};  ! refers to scale factors

          variable$number  ! label for the threshold of a variable

          variable#number   ! label for a latent class

MODEL label:  ! refers to a group-specific model in multigroup analysis

          %BETWEEN%  ! refers to the between part of two-level model

          %WITHIN%  ! refers to within part of two-level model

          %OVERALL%  ! refers to overall model in latent mixture model

          %class label%  ! refers to class-specific model in latent mixture model

          MONTECARLO:  ! refers to data generation model

          MISSING:  ! refers to missing data generation model in Monte Carlo

Keywords

There are four keywords in the MODEL command that are used to specify relationships among variables, ON, BY, WITH, PWITH. There is some flexibility in the use of these commands. You can specify individual relationships one at a time or in groups. The two commands

y1 ON x1 x2 x3;           y2 ON x1 x2 x3;

are equivalent to the command

y1 y2 ON x1 x2 x3;

ON stands for "regressed on" and specifies a regular regression relationship.

BY stands for "measured by" and is used to declare a latent variable (on the left of the BY statement) and specify its relationship to indicator variables (on the right of the BY statement).  For example, 

LV1 BY x1-x3; 

would  create a new variable LV1 (this would be a new name not in the list of variables) and declare x1, x2, and x3 to be indictors for that latent.

WITH stands for "correlates with". This is useful for specifying error correlations and other correlations not allowed in the defaults. If you put a list of variables on either left or right or both, all pairwise combinations are specified.

x1 x2 WITH x3 x4;

implies -   x1 WITH x3;  x1 WITH x4;  x2 WITH x3;  x2 WITH x4; 

PWITH stands for "paired with" and is used to pair the list of variables on the left side with those on the right, thus, the number of variables on each side must be equal. 

x1 x2 PWITH x3 x4;

implies -   x1 WITH x3;  x2 WITH x4; 

Assigning Starting Values and Freeing Parameters for Estimation

The asterisk (*) is used for two special purposes. One use to assign a starting value, in which case the asterisk will be followed by a numeric value, e.g.,

y1*0.25;

This statement actually means two separate things; first, it means the value of y1 is to be estimated (simply by having the asterisk follow y1, second, it means the starting value for estimation should be 0.25. This statment further implies that there might exist cases where asterisks are used, but without a starting value. When would this be the case? When the default causes a value to be set to a fixed value, which happens with latent variables having multiple indicators. Consider the statement above that demonstrates the BY command,

LV1 BY x1-x3;

What I didn't tell you when I first introduced this command example is that a default procedure in Mplus is to fix the value of the loading for the first indicator to 1.0. The reason for this procedure will be explained later when we talk about latent variable models. For now, we need to recognize that if we wanted to allow the loading of x1 on LV1 to be freely estimated, we would have to ask for that to happen using the command

LV1 BY x1* x2 x3;

"Fixing" (i.e., specifying) Values for Parameters

The phrase "fixing values" or "fixing parameters" is SEM lingo for specifying an invariant value for a parameter. For example, imagine you had a situation as shown here

One of Mplus's other defaults is that it automatically allows all exogenous variables to freely intercorrelate (in this case, the two exogenous variables are x1 and x2). If we wanted to set the correlation/covariance between x1 and x2 to zero, we would have to add the command

x1WITHx2@0;

The other common use for the @ keyword is to specify the scale for a latent variable using other than the first indicator listed. To do this, we would say

LV1 BY x1* x2 x3@1;

Constraining Values to be Equal

Parameter values can be constrained to be equal by placing the same number in parentheses following the parameters to be held equal. For example, if we had the two commands 

y1 ON x1   (1);

y2 ON x2   (1);

this would mean that we wished to test whether the regressions y1 = x1 and y2 = x1 had the same slope (actually, we are specifying that they have equal slopes, the test as to whether this is true would involve assessing whether such a model had adequate fit or not). 

Variances, Means, Thresholds, Scale Factors, & Latent Classes

If we include a statement under the MODEL: command like

x1@1.0;

Then we are specifying that the variance of the variable x1 is being set equal to 1.0. If we were dealing with a response variable, then we would be referring to its error variance instead of its variance. On the other hand, if we were to specify

[x1]@1.0;

then we would be specifying that the mean of the variable x1 was equal to 1.0. If we were dealing with a response variable, then the brackets signify that we are dealing with the intercept.

If we were dealing with a categorical response variable, what would be contained in brackets would be the thresholds. For a categorical variable with three levels, we could specify the values for its thresholds by stating

[y1$1]@0.33; 

[y1$2]@0.67;

Finally, curly brackets {} signify that we are specifying scale factors. This and the specification of labels for thresholds and latent classses are topics I will leave to the user's guide for now. Same holds for labeling model parts for more complex models, such as multi-level models.

OUTPUT Command:

Command Options for OUTPUT:

OUTPUT:

          SAMPSTAT;  ! requests sample statistics; usually needed

          MODINDICES(cutoff); ! requests modification indices; use (3.84)

          STANDARIZED; ! requests standardized values; usually desirable

          RESIDUAL; ! requests residuals for model evaluation

          CINTERVAL; ! 95% and 99% confidence intervals for parameters

          H1SE; ! standard errors for H1 model

          H1TECH3; ! estimated covariance and correlation matrices for H1 model

          PATTERNS; ! summary of missing data patterns

          FSCOEFFICIENT; ! factor score coefficients

          FSDETERMINACY; ! factor score determinancy value for each factor

          TECH1-9; ! requests different technical outputs

Command Options

Some of the above options are obvious and some will be cryptic for many. It is definitely worth reading up on the different technical outputs. As always, you can read more in the Mplus User's Guide on all this.

SAVEDATA Command:

Command Options for SAVEDATA:

SAVEDATA: 

          FILE IS                              name of file;

          FILE (SAMPLE) IS          name of file;

          FILE (SIGB) IS                name of file;

          FILE (RESULTS) IS        name of file;

          FILE (TECH3) IS             name of file;

          FILE (TECH4) IS             name of file;

          FORMAT IS                       format statement;

          TYPE (SAMPLE) IS        COVARIANCE; or CORRELATION;

          TYPE (SIGB) IS               COVARIANCE; or CORRELATION;

          RECORDLENGTH IS      characters per record;

          SAVE =                                  FSCORES; CPROBABILITIES;

Options

FILE

This option names the ascii file to be written. If the SAVEDATA command is used, the FILE option is required.

There are six different forms of the command option statement, depending on the type of data to be saved:

FILE IS statement saves individual-level analysis data.

FILE (SAMPLE) IS saves the sample statistics.

FILE (SIGB) IS saves the estimated sigma between matrix in two-level analysis.

FILE (RESULTS) IS saves the results of an analysis.

FILE (TECH3) IS saves the TECH3 output.

FILE (TECH4) IS saves the TECH4 output.

The simplest FORMAT IS statement to use is FORMAT IS FREE; See the user's manual for more information.

The TYPE option statement specifies the type of data to be saved.

The RECORDLENGTH option is used to specify the number of characters per record for the file. The default and maximum value is 1000;

SAVE = FSCORES; means the factor scores are saved along with the other analysis variables.

SAVE = CPROBABILITIES; saves the class probabilities. Must be used in conjunction with the TYPE=MIXTURE analysis command.

Other savedata options have been added in the latest version of Mplus. See user's guide.

PLOT Command:

Command Options for PLOT:

The PLOT command is used to obtain graphical displays of histograms, scatter plots, and some of the result outputs.

PLOT:

TYPE IS                       PLOT1;

                                      PLOT2;

                                      PLOT3;

SERIES IS                   list of variables in a series;

OUTLIERS ARE       list of options for detecting outliers;

The graphics available with the different PLOT options are as follows:

PLOT1 - histograms, scatterplots, sample means, sample proportions, individual values.

PLOT2 - sample means, sample proportions, estimated means, estimated probabilities, estimated and sample means, adjusted means, item characteristic curves, information curves, eigen values for EFA, mixture distributions.

PLOT3 - all the above plus a few other things.

SERIES option is for variables where the plots will connect points with lines. For growth models, the plot connects the repeated measures for each observation.

Note that to view the graphs, you go up to the Graph menu item in the command window.

MONTECARLO Command:

Command Options for MONTECARLO:

This command is used for methodological studies, for example, to evaluate the performance of different estimators under various conditions. The MONTECARLO command is used in conjunction with the ANALYSIS, MODEL, and OUTPUT commands to generate, transform, analyze, and summarize results from data for Monte Carlo simulation studies.

For continuous variables, the MONTECARLO command creates multivariate normal data by repeatedly sampling from a model specified by you, the user. The command can also be used to generate data for categorical variables. When generating a montecarlo data set, an external dataset is required that contains the means plus the variances and covariances for all variables.

MONTECARLO:

     FILE =                                  name of file containing info needed for analysis;

     NOBSERVATIONS =    number of observations;

     NAMES =                            names of variables;

     NGROUPS =                      number of groups;

     NREPS =                             number of reps;

     SEED =                                 random seed for data generation;

     CUTPOINTS =                 thresholds to be used for categorization;

     CATEGORICAL =            number of thresholds to be used;

     SAVE =                                 name of file where data from first rep is stored;

     NCLASSES =                      number of classes in analysis model;

     GCLASSES =                       number of classes in data generation model;

     PATMISS =                         missing data patterns;

     PATPROBS =                     proportion for each missing data pattern;

     MISSING =                         variables that have missing data;

     RESULTS =                        name of file in which results are stored;

Note that the OUTPUT command has two options that provide information specific to Monte Carlo studies, TECH1 and TECH9. TECH1 shows the model structure and the parameter numbers and values used in the analysis. TECH9 is for when there are categorical variables.

Also note that a key component of performing a Monte Carlo study is the specification of the MODEL statement. The MODEL statement is used to provide the true population values for the simulation.

MODEL MONTECARLO:

     y1 ON x1@0.75; ! sets parameter value for paths

     y2 ON x1@0.3 x2@0.1 y1@0.45; ! sets parameter values for paths

     [x1-x3@1.0]; ! sets the variances of the predictors to 1.0

